

Using Variables

- **Variables** tell the computer to *remember* something.
- **Variables** can change (vary).
- **Variables** are like boxes or containers to hold things.
- There are different size boxes to hold different things.
- Variable names are usually written in camelCase

Try these code snippets to see how variables work.

```
void setup() {
  size(800, 600);
  background(200);
  circle(400, 300, 200);
  // This draws a circle in the middle of the screen
}
```

What if I want to change my screen size? I'll need to change my circle's position too.

Now try this:

```
void setup() {
  size(800, 600);
  background(200);
  circle(width/2, height/2, 200); // "width divided by 2", "height divided by 2"
  // This draws a circle in the middle of the screen
}
```

This does the same thing, but now my code will automatically move the circle to the middle if I change my screen size.

Things To Try

1. Change the screen size a few times and see how the circles stays centered.

Animation with Variables

Variables let us animate our sketches. Try this code.

```
int x = 1; // create an "integer" variable named "x" and set its value to "1"

void setup() {
  size(800, 600);
}

void draw() {
  background(200);
  circle(x, 300, 30); // draw the circle at position x
  x = x + 1; // increase x by 1 each loop
}
```

Congratulations! You just created your first variable.

Things To Try

1. Change the ball's speed.
2. Make the ball go from right to left.
3. Can you make the ball fall from the top down, instead of moving from left to right?

Pre-Made Variables

Some variables come pre-made in Processing. You can just use these.

- mouseX = the "x" position of your mouse
- mouseY = the "y" position of your mouse
- width = whatever you set as the screen width
- height = whatever you set as the screen height

Making Your Own Variables

You can also make your own **variables**. Each variable creates a "box" in the computer's memory to hold some information. To create a **variable** you need to tell the computer three things.

1. What kind of box you want to create (will it hold numbers, words, etc.)
2. A name for your box
3. Information (a value) to put in the box

There are four basic kinds of boxes you can make

1. **int** = Whole numbers (integers), like 1, 42, or -15
2. **float** = Decimal numbers (floating point), like 3.14, 15.99, or -100.4
3. **String** = Words (a String of characters), like "hello", or "this is fun"
4. **boolean** True/False (called a boolean), like true, or false

For now, we'll just use the first kind of box—Integers.

Try this code:

```
int x = 0; // "x" position of the circle
int y = 0; // "y" position of the circle
int w = 10; // "w" is for the width of our circle

void setup() {
  size(800, 600);
}

void draw() {
  background(200);
  circle(x, y, w); // draw the circle at position x
  x = x + 1; // increase x by 1 each loop
  y = y + 1; // increase y by 1 each loop
}
```

Things To Try

1. Change the values of the x, y, and w variables.
2. Rename the variables to a, b, and c. (Hint: if you highlight a variable and right click, you get a menu with the option to rename. This will automatically replace every time you used that variable with the new name. Otherwise, you need to change them all yourself.)
3. Make a bulls-eye with three or more circles. Color them red, white, and black. Use variables to center the bulls-eye to the screen.
4. Challenge: make the bulls-eye follow your mouse around the screen!